
    Antenna

The tiny framework to simplify work with events, based on GenStage
Objective
Antenna is a mixture of Phoenix.PubSub
and :gen_event functionality
with some batteries included.
It implements back-pressure on top of GenStage, is fully conformant with
OTP Design Principles. and
is distributed out of the box.
Antenna supports both asynchronous and synchronous events. While the most preferrable way
would be to stay fully async with Antenna.event/3, one still might propagate the event
synchronously with Antenna.sync_event/3 and collect all the responses from all the handlers.
One can have as many isolated Antennas as necessary, distinguished by Antenna.t:id/0.
The workflow looks like shown below.
Flow
                        ┌───────────────┐                                               
                        │               │       Antenna.event(AntID, :tag1, %{foo: 42}) 
                        │  Broadcaster  │───────────────────────────────────────────────
                        │               │                                               
                        └───────────────┘                                               
                              ──/──                                                     
                        ────── /   ─────                                                
                  ──────      /         ──────                                          
            ──────           /                ─────                                     
        ────                /                      ───                                  
┌────────────────┐  ┌────────────────┐       ┌────────────────┐                         
│                │  │                │       │                │                         
│ Consumer@node1 │  │ Consumer@node1 │   …   │ Consumer@nodeN │                         
│                │  │                │       │                │                         
└────────────────┘  └────────────────┘       └────────────────┘                         
        ·                   ·                         ·                                 
      /   \               /   \                     /   \                               
    /       \           /       \                 /       \                             
  ·   mine?   ·       ·   mine?   ·             ·   mine?   ·                           
    \       /           \       /                 \       /                             
      \   /               \   /                     \   /                               
        ·                   ·                         ·                                 
        │                   │                         │                                 
        │                   │                         │                                 
      ─────                 │                       ─────                               
                            │                                                           
                            │                                                           
                     ┌──────────────┐                                                   
                     │              │             if (match?), do: call_handlers(event) 
                     │   matchers   │───────────────────────────────────────────────────
                     │              │                                                   
                     └──────────────┘                                                   
ASCII representation.
Usage Example
The consumer of this library is supposed to declare one or more matchers, subscribing to one
or more channels, and then call Antenna.event/2 to propagate the event.
Installation
def deps do
  [
    {:antenna, "~> 0.2"}
  ]
end
Is it of any good?
Sure, it is.
Documentation


  

    Smart Eventing

Antenna is an elegant event handling framework that combines the best features of Phoenix.PubSub and gen_event with sophisticated pattern matching and back-pressure control. Built on GenStage and fully distributed by design, it offers a powerful yet simple way to handle events across your Elixir applications.
Key Features
1. Smart Pattern Matching
Unlike traditional pub/sub systems, Antenna provides Elixir-native pattern matching capabilities:
# Match on complex patterns with guards
Antenna.match(MyApp.Antenna, 
  {:temperature, celsius, location} when celsius > 30,
  fn channel, event ->
    Logger.warning("High temperature alert: #{inspect(event)} on #{channel}")
  end,
  channels: [:sensors])
2. Built for Distribution
	Automatic event propagation across connected nodes
	Node-local pattern matching for efficiency
	Built-in back-pressure through GenStage
	Process group-based channel management

# Events automatically propagate to all connected nodes
Node.connect(:"node2@host2")
Antenna.event(MyApp.Antenna, [:alerts], {:system_status, :degraded})
3. Flexible Event Handling
Support for both asynchronous and synchronous event processing:
# Async event dispatch
Antenna.event(MyApp.Antenna, [:logs], {:user_action, user_id, :login})

# Sync event with collected responses
responses = Antenna.sync_event(MyApp.Antenna, [:auth], {:verify_token, token})
4. Channel-based Organization
	Multiple isolated Antenna instances
	Channel-based event routing
	Support for wildcard subscriptions
	Dynamic channel subscription management

# Different antennas for different subsystems
Antenna.event(:metrics, [:system], %{cpu: 80, memory: 70})
Antenna.event(:business, [:orders], %{order_id: "123", status: :completed})
5. OTP-Compliant Design
	Follows OTP design principles
	Supervisor-based process management
	Clean process isolation
	Proper error handling and recovery

Real-world Applications
Event-Driven Microservices
# Service A: Order Processing
Antenna.match(OrderSystem, 
  %{event: :order_created, total: total} when total > 1000,
  fn _, event ->
    HighValueOrderProcessor.handle(event)
  end,
  channels: [:orders])

# Service B: Notification System
Antenna.match(NotificationSystem,
  {:order_status, order_id, :completed},
  fn _, event ->
    CustomerNotifier.order_completed(order_id)
  end,
  channels: [:order_updates])
IoT Data Processing
# Device data handling with pattern matching
Antenna.match(IoTSystem,
  {:sensor_data, device_id, readings} when readings.temperature > 90,
  fn channel, event ->
    DeviceAlertHandler.process_alert(channel, event)
  end,
  channels: [:device_telemetry])
Distributed Systems Monitoring
# System metrics collection
Antenna.match(Monitoring,
  {:metric, node_name, metric, value} when value > threshold(),
  fn _, event ->
    MetricsAggregator.record_threshold_breach(event)
  end,
  channels: [:system_metrics])
Getting Started
	Add Antenna to your dependencies:

def deps do
  [{:antenna, "~> 0.2"}]
end
	Add it to your supervision tree:

defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      {Antenna, name: MyApp.EventSystem}
    ]
    
    Supervisor.start_link(children, strategy: :one_for_one)
  end
end
	Set up your event handlers:

# Define matchers for your events
Antenna.match(MyApp.EventSystem,
  {:user_event, user_id, action},
  fn channel, event ->
    MyApp.EventTracker.track(channel, event)
  end,
  channels: [:user_activity])

# Start sending events
Antenna.event(MyApp.EventSystem, [:user_activity], 
  {:user_event, "user123", :login})
Why Choose Antenna?
	Native Pattern Matching: Leverage Elixir's powerful pattern matching for event handling
	Distributed by Design: Built for distributed systems with automatic node synchronization
	Back-pressure Control: Built on GenStage for controlled event flow
	Flexible Configuration: Multiple isolated instances and channel-based routing
	Production Ready: OTP-compliant with proper supervision and error handling

Internal Architecture
Event Flow
Antenna processes events through a distributed pipeline:
                        ┌───────────────┐                                               
                        │               │       Antenna.event(AntID, :tag1, %{foo: 42}) 
                        │  Broadcaster  │───────────────────────────────────────────────
                        │               │                                               
                        └───────────────┘                                               
                              ──/──                                                     
                        ────── /   ─────                                                
                  ──────      /         ──────                                          
            ──────           /                ─────                                     
        ────                /                      ───                                  
┌────────────────┐  ┌────────────────┐       ┌────────────────┐                         
│                │  │                │       │                │                         
│ Consumer@node1 │  │ Consumer@node1 │   …   │ Consumer@nodeN │                         
│                │  │                │       │                │                         
└────────────────┘  └────────────────┘       └────────────────┘                         
        ·                   ·                         ·                                 
      /   \               /   \                     /   \                               
    /       \           /       \                 /       \                             
  ·   mine?   ·       ·   mine?   ·             ·   mine?   ·                           
    \       /           \       /                 \       /                             
      \   /               \   /                     \   /                               
        ·                   ·                         ·                                 
        │                   │                         │                                 
        │                   │                         │                                 
      ─────                 │                       ─────                               
                            │                                                           
                            │                                                           
                     ┌──────────────┐                                                   
                     │              │             if (match?), do: call_handlers(event) 
                     │   matchers   │───────────────────────────────────────────────────
                     │              │                                                   
                     └──────────────┘                                                   
Components
	Broadcaster (GenStage Producer)
	Distributes events to all nodes
	Implements back-pressure control
	Handles sync/async event dispatch


	Process Groups
	Channel-based process groups for event routing
	Dynamic subscriber management
	Node-local event matching


	Pattern Matchers
	Handle complex pattern matching
	Support guards and conditions
	Route events to appropriate handlers


	Guard
	Supervises channel subscriptions
	Manages handler lifecycle
	Ensures clean process termination



Back-pressure
Antenna uses GenStage to implement back-pressure:
	Events are dispatched based on consumer demand
	System stays responsive under load
	Memory usage remains stable
	Event ordering is preserved

Distribution
Built-in distribution capabilities:
	Automatic node discovery
	Transparent event propagation
	Node-local pattern matching
	Efficient event routing

For a detailed implementation guide and more examples, visit the official documentation.
Configuration
Antenna can be configured in your config.exs:
config :antenna,
  id: MyApp.Antenna,           # Optional custom identifier
  distributed: true,           # Enable distributed mode (default: true)
  sync_timeout: 5_000          # Default timeout for sync_event/4
Best Practices
1. Channel Organization
	Use atoms for channel names when possible
	Group related events under common channels
	Consider using hierarchical channel naming:# Good channel organization
Antenna.event(MyApp.Antenna, [:users_created], event)
Antenna.event(MyApp.Antenna, [:users_updated], event)


2. Pattern Matching
	Use specific patterns to avoid unnecessary matches:# Good - specific pattern
Antenna.match(MyApp.Antenna, {:user_created, user_id, meta}, handler)

# Less efficient - overly broad pattern
Antenna.match(MyApp.Antenna, {_action, _id, _}, handler)

	Include guards for more precise matching
	Consider the order of pattern matches when using multiple matchers

3. Handler Design
	Keep handlers focused and single-purpose
	Use sync_event/4 only when you need responses
	Consider timeouts for sync operations:# Set appropriate timeout for sync operations
responses = Antenna.sync_event(MyApp.Antenna, [:auth], event, 10_000)

	Handle errors within handlers to prevent cascade failures

4. Performance
	Use async events (event/3) by default
	Keep handler processing time minimal
	Consider using separate processes for long-running operations:Antenna.match(MyApp.Antenna, event_pattern, fn channel, event ->
  # Spawn long-running operations
  Task.start(fn -> LongRunningProcessor.process(event) end)
end)

	Monitor matcher and handler counts

5. Testing
	Test matchers with various event patterns:test "matches high temperature events" do
  Antenna.match(TestAntenna, 
    {:temperature, val, _} when val > 30,
    self(),
    channels: [:test])
    
  Antenna.event(TestAntenna, [:test], {:temperature, 35, :room})
  assert_receive {:antenna_event, :test, {:temperature, 35, :room}}
end

	Test handler behavior with both valid and invalid events
	Test distributed scenarios
	Use ExUnit's async: true when possible

Conclusion
Antenna provides a robust foundation for building distributed event-driven systems in Elixir. Whether you're building microservices, processing IoT data, or managing system events, Antenna offers the tools you need with the elegance and reliability you expect from Elixir applications.


  

    DIY Message Broker

In Erlang (and Elixir), I always missed a way to organize “streaming” message exchange, similar to what a Message Broker provides. Normal developers accept the constraints imposed by their frameworks: Phoenix has PubSub, OTP has :gen_event, and Elixir had GenEvent, deprecated even before it was born.
None of these suited me. PubSub is powerful, but its asynchronicity is nailed down, and sometimes you need to call subscribers synchronously (I know you can send your pid with the message and wait for responses, but this ad-hoc approach unfortunately doesn’t help when the number of subscribers is unknown). :gen_event is almost what I need, but its design clearly exposes side-effect handlers, making convenient filtering of incoming message streams turn into spaghetti. In general, as always: if you don’t do it yourself, no one will.
Thus, the Antenna library was born, providing all the features usually achieved by adding an extra message broker dependency. My requirements for it were as follows:
	Back pressure support out of the box (a million events sent at once should not overwhelm handlers, and adding new nodes to the cluster should transparently offload them)
	Channels (topics, tags) where you can send an event that will be delivered to all registered handlers
	Handlers (besides channel subscriptions) should be able to transparently filter the incoming message stream using the full power of Erlang’s pattern matching
	Both handlers and subscriptions should successfully recover after a crash
	The message sender can request synchronous execution of handlers, blocking the calling process until all responses are received (like RPC)
	The library should provide convenient primitives for testing its usage

Anyone who has used RabbitMQ would have noticed where I stole my ideas from easily.
Pseudocode
For a long time, I’ve used a kind of TDD with a hint of DDD, or vice versa—I’m not strong in buzzword abbreviations. In general, I start designing any piece of code by calling its (yet non-existent) API. First, I make sure that calling my code is easy and pleasant, and I’ll figure out the implementation later. So I sketched something like this:
{:ok, pid} = match({:tag_answer, _}, self())
subscribe(:chan, pid)

event(:chan, {:tag_answer, 42})

receive do
  {:event, :chan, {:tag_answer, 42}} -> :ok
after
  1_000 -> :error
end
The first two lines are handler registration and subscription. The next is sending a message. The final block checks that the message was received. These three pieces will usually be in different, unrelated parts of the code.
The first thing I noticed is that the first two lines almost always go together, so it makes sense to accept a list of channels for subscription right in the “matcher” definition (without canceling the subscribe/2 function, in case someone wants to subscribe to a channel later).
I also noticed that there can be many handlers, why not.
Thus, the unique identifier for a matcher remains the match itself. It can be quite tricky, so I chose its textual representation as the id (the string "{:tag_answer, _}" for the example above). I don’t really like this solution, but I don’t have a better one (yet). At least, inspecting matchers is much easier this way than, for example, using something hash-like.
So, we get a many-to-many relationship: many handlers for one match, which can receive messages from many channels. For example, we can subscribe to all messages of the form {:error, _} from all channels and assign two handlers: a logger will print to the console, telemetry will send something to Grafana, or wherever it’s customary to send everything.
Main Architecture
Back pressure in Elixir implies using the GenStage library. I already implemented my Throttler with it, now it’s the broker’s turn. We remember about horizontal scaling—which means several consumers, at least one per node. Each consumer will send messages to specified channels, matchers will check if they need to bother (whether the message matches), and if so, call the handlers. Sounds reasonable.
I spread consumers across nodes using unnamed processes managed by DistributedSupervisor, the broadcaster too (except this process is named, so there’s only one per cluster, and it will be restarted on another node if the current one disappears).
Each matcher is also a process, which stores a list of handlers and the match itself in its state. Here I faced my first architectural dilemma: how to store the match? {:foo, _} can’t just be stored as is; such code is only allowed as LHO in direct pattern matching calls, and storing its AST isn’t an option, because then you can’t insert it into a match. So, I decided to generate a matcher function (anyway, implementing match/4 is only possible as a macro):
quote generated: true, location: :keep do
  matcher = fn
    unquote(match) -> true
    _ -> false
  end
  …
end
Great. Now, in the matcher process itself, there’s a callback like this:
  @impl GenServer
  @doc false
  def handle_cast({:handle_event, channel, event}, state) do
    if state.matcher.(event) do
      Enum.each(state.handlers, fn
        handler when is_function(handler, 1) -> handler.(event)
        handler when is_function(handler, 2) -> handler.(channel, event)
        process -> send(process, {:antenna_event, channel, event})
      end)
    end

    {:noreply, state}
  end
And we’re ready to send it messages from consumers (matchers are also managed by DistributedSupervisor, i.e., evenly spread across the cluster).
At this point, I said git init because the MVP was shaping up.
Synchronous Call
Up to this point, everything was pretty trivial. But how do you organize a synchronous call when all messages go through seven circles of hell (broadcast, consumers, matchers-all on who-knows-which nodes in the cluster)?
Have you ever wondered why the signature of GenServer’s asynchronous callback is of arity 2, while the synchronous one is of arity 3? By the way, this is one of my favorite interview questions: it immediately becomes clear whether you’re dealing with a dreamy reactive hipster or a nerdy freaking sociopath.
The second argument in the handle_call/3 callback is the process identifier (without loss of generality) waiting for a synchronous response. And instead of the tuple {:reply, result, state} from this callback, you can return {:noreply, new_state}, and later send a synchronous reply directly to the calling process using GenServer.reply/2. If you didn’t know this, have a sip of whiskey for me: this is the hammer that makes everything around you nails.
GenStage, in turn, also exports reply/2. So now I just need to pass from through all consumers and matchers, and then deep inside write something like:
Enum.each(results, fn
  {nil, _} -> :ok
  {from, results} -> GenStage.reply(from, results)
end)
If it was an asynchronous call-there’s no from, and we do nothing. But if it exists-we send back the accumulated results of all handler calls (and let it choke on them).
That seems to be all I wanted to share today. The link to the library, its source code, and tests are above.
Happy brokering!

Antenna is a mixture of Phoenix.PubSub
  and :gen_event functionality
  with some batteries included.
It implements back-pressure on top of GenStage, is fully conformant with
  OTP Design Principles. and
  is distributed out of the box.
Antenna supports both asynchronous and synchronous events. While the most preferrable way
  would be to stay fully async with Antenna.event/3, one still might propagate the event
  synchronously with Antenna.sync_event/3 and collect all the responses from all the handlers.
One can have as many isolated Antennas as necessary, distinguished by Antenna.t:id/0.
Getting Started
To start using Antenna, add it to your dependencies in mix.exs:
def deps do
  [
    {:antenna, "~> 0.3.0"}
  ]
end
Then add it to your supervision tree:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      Antenna
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Configuration
Antenna can be configured in your config.exs:
config :antenna,
  id: MyApp.Antenna,           # Optional custom identifier
  distributed: true,           # Enable distributed mode (default: true)
  sync_timeout: 5_000          # Default timeout for sync_event/4
Sequence Diagram
sequenceDiagram
  Consumer->>+Broadcaster: sync_event(channel, event)
  Consumer->>+Broadcaster: event(channel, event) 
  Broadcaster-->>+Consumer@Node1: event
  Broadcaster-->>+Consumer@Node2: event
  Broadcaster-->>+Consumer@NodeN: event
  Consumer@Node1-->>-NoOp: mine?
  Consumer@NodeN-->>-NoOp: mine?
  Consumer@Node2-->>+Matchers: event
  Matchers-->>+Handlers: handle_match/2
  Matchers-->>+Handlers: handle_match/2
  Handlers->>-Consumer: response(to: sync_event)
ASCII representation.
Common Use Cases
Basic Event Handling
# Define a matcher for specific events
Antenna.match(MyApp.Antenna, {:user_registered, user_id, _meta}, fn channel, event ->
  # Handle the event
  {:user_registered, user_id, meta} = event
  Logger.info("New user registered: #{user_id} on channel #{channel}")
  MyApp.UserNotifier.welcome(user_id)
end, channels: [:user_events])

# Send events
Antenna.event(MyApp.Antenna, [:user_events], {:user_registered, "user123", %{source: "web"}})
Error Handling
# Match on error events
Antenna.match(MyApp.Antenna, {:error, reason, _context} = event, fn channel, event ->
  # Log and handle errors
  Logger.error("Error on channel #{channel}: #{inspect(event)}")
  MyApp.ErrorTracker.track(event)
end, channels: [:errors])

# Send error events
Antenna.event(MyApp.Antenna, [:errors], {:error, :validation_failed, %{user_id: 123}})
Distributed Setup
Antenna automatically handles distribution when nodes are connected:
# On node1@host1
Node.connect(:"node2@host2")
Antenna.match(MyApp.Antenna, {:replicate, data}, fn _, {:replicate, data} ->
  MyApp.Storage.replicate(data)
end, channels: [:replication])

# On node2@host2
Antenna.event(MyApp.Antenna, [:replication], {:replicate, some_data})
# The event will be processed on both nodes
Custom Matchers
You can use pattern matching and guards in your matchers:
# Match on specific patterns with guards
Antenna.match(MyApp.Antenna, 
  {:temperature, celsius, _} when celsius > 30,
  fn _, {:temperature, c, location} ->
    Logger.warning("High temperature (#{c}°C) detected at #{location}")
  end,
  channels: [:sensors])

# Match on map patterns
Antenna.match(MyApp.Antenna,
  %{event: :payment, amount: amount} when amount > 1000,
  fn _, event ->
    MyApp.LargePaymentProcessor.handle(event)
  end,
  channels: [:payments])
Usage Example
The consumer of this library is supposed to declare one or more matchers, subscribing to one
  or more channels, and then call Antenna.event/2 to propagate the event.
assert {:ok, _pid, "{:tag_1, a, _} when is_nil(a)"} =
  Antenna.match(Antenna, {:tag_1, a, _} when is_nil(a), self(), channels: [:chan_1])

assert :ok = Antenna.event(Antenna, [:chan_1], {:tag_1, nil, 42})
assert_receive {:antenna_event, :chan_1, {:tag_1, nil, 42}}
Architecture
Antenna uses a distributed pub/sub architecture with the following components:
	Broadcaster - Handles event distribution across nodes
	Matchers - Pattern match on events and delegate to handlers
	Guard - Manages channel subscriptions and handler registration
	PubSub - Implements the core pub/sub functionality

Events flow through the system as follows:
	Client calls event/3 or sync_event/4
	Broadcaster distributes the event to all nodes
	Matchers on each node pattern match against the event
	Matching handlers process the event
	For sync_event, responses are collected and returned

Best Practices
	Channel Organization
	Use atoms for channel names
	Group related events under common channels
	Consider using hierarchical channel names (e.g., :users_created, :users_updated)


	Pattern Matching
	Use specific patterns to avoid unnecessary pattern matches
	Include guards for more precise matching
	Consider the order of pattern matches when using multiple matchers


	Handler Design
	Keep handlers focused and single-purpose
	Use sync_event/4 when you need responses
	Consider timeouts for sync operations
	Handle errors within handlers to prevent cascade failures


	Performance
	Use async events (event/3) when possible
	Keep handler processing time minimal
	Consider using separate processes for long-running operations
	Monitor matcher and handler counts


	Testing
	Test matchers with various event patterns
	Verify handler behavior
	Test distributed scenarios
	Use ExUnit's async: true when possible




      


      
        Summary


  
    Functions (Client)
  


    
      
        event(id \\ Antenna, channels, event)

      


        Sends an event to all the associated matchers through channels.



    


    
      
        sync_event(id \\ Antenna, channels, event, timeout \\ 5000, extra_timeout \\ 100)

      


        Sends an event to all the associated matchers through channels synchronously,
collecting and returning responses from all handlers.



    





  
    Functions (Setup)
  


    
      
        child_spec(init_arg)

      


        Returns a specification to start this module under a supervisor.



    


    
      
        handle(id \\ Antenna, handlers, pid)

      


        Adds a handler to the matcher process specified by pid



    


    
      
        match(id \\ Antenna, match, handlers, opts \\ [])

      


        Declares a matcher for tagged events.



    


    
      
        start_link(init_arg \\ [])

      


        Starts the Antenna matcher tree for the id given. 



    


    
      
        subscribe(id \\ Antenna, channels, pid)

      


        Subscribes a matcher process specified by pid to a channel(s)



    


    
      
        unhandle(id \\ Antenna, handlers, pid)

      


        Removes a handler from the matcher process specified by pid



    


    
      
        unmatch(id \\ Antenna, match)

      


        Undeclares a matcher for tagged events previously declared with Antenna.match/4.



    


    
      
        unsubscribe(id \\ Antenna, channels, pid)

      


        Unsubscribes a previously subscribed matcher process specified by pid from the channel(s)



    





  
    Functions (Internals)
  


    
      
        registered_matchers(id)

      


        Returns a map of matches to matchers



    





  
    Types
  


    
      
        channel()

      


        The identifier of the channel, messages can be sent to, preferrably atom()



    


    
      
        event()

      


        The event being sent to the listeners



    


    
      
        handler()

      


        The actual handler to be associated with an event(s). It might be either a function
  or a process id, in which case the message of a following shape will be sent to it.



    


    
      
        id()

      


        The identifier of the isolated Antenna



    


    
      
        matcher()

      


        The matcher to be used for the event, e.g. {:tag_1, a, _} when is_nil(a)



    


    
      
        opts()

      


        The options to be passed to Antenna.match/4 and Antenna.unmatch/2



    





      


      
        Functions (Client)

        


    

  
    
      
    
    
      event(id \\ Antenna, channels, event)



        
          
        

    

  


  

      

          @spec event(id :: id(), channels :: channel() | [channel()], event :: event()) :: :ok


      


Sends an event to all the associated matchers through channels.
The special :* might be specified as channels, then the event
  will be sent to all the registered channels.
If one wants to collect results of all the registered event handlers,
  they should look at sync_event/3 instead.
Examples
Basic Usage
# Send event to a single channel
Antenna.event(MyApp.Antenna, :user_events, {:user_logged_in, "user123"})

# Send event to multiple channels
Antenna.event(MyApp.Antenna, [:logs, :metrics], {:api_call, "/users", 200, 45})

# Send event to all channels
Antenna.event(MyApp.Antenna, :*, {:system_notification, :service_starting})
Custom Antenna ID
# Using a specific Antenna instance
Antenna.event(MyBackgroundJobs.Antenna, :jobs, {:job_completed, "job-123"})

  



    

    

    

  
    
      
    
    
      sync_event(id \\ Antenna, channels, event, timeout \\ 5000, extra_timeout \\ 100)



        
          
        

    

  


  

      

          @spec sync_event(
  id :: id(),
  channels :: channel() | [channel()],
  event :: event(),
  timeout :: timeout(),
  extra_timeout :: timeout()
) :: [term()]


      


Sends an event to all the associated matchers through channels synchronously,
collecting and returning responses from all handlers.
The special :* might be specified as channels, then the event
will be sent to all the registered channels.
Unlike event/3, this function waits for all handlers to process the event
and returns their responses.
Parameters
	id - The Antenna instance identifier (optional, defaults to configured value)
	channels - Single channel or list of channels to send the event to
	event - The event to be sent
	timeout - Maximum time to wait for responses (default: 5000ms)

Examples
Basic Usage
# Send sync event and collect responses
results = Antenna.sync_event(MyApp.Antenna, :user_events, {:verify_user, "user123"})

# Send to multiple channels with custom timeout
results = Antenna.sync_event(
  MyApp.Antenna,
  [:validations, :security],
  {:user_action, "user123", :delete_account},
  10_000
)
Response Format
The function returns a list of handler responses, each wrapped in a tuple
indicating whether the pattern matched and including the handler's result:
[
  {:match, %{
    match: "pattern_string",
    channel: :channel_name,
    results: [handler_result]
  }},
  {:no_match, :channel_name}
]

  


        

      

      
        Functions (Setup)

        


  
    
      
    
    
      child_spec(init_arg)



        
          
        

    

  


  

Returns a specification to start this module under a supervisor.
See Supervisor.

  



    

  
    
      
    
    
      handle(id \\ Antenna, handlers, pid)



        
          
        

    

  


  

      

          @spec handle(id :: id(), handlers :: handler() | [handler()] | MapSet.t(), pid()) ::
  :ok


      


Adds a handler to the matcher process specified by pid

  



    

    

  
    
      
    
    
      match(id \\ Antenna, match, handlers, opts \\ [])


        (macro)


        
          
        

    

  


  

Declares a matcher for tagged events.
This function sets up a pattern matcher to handle specific events on the given channels.
When an event matching the pattern is sent to any of the channels, the provided handler
will be invoked.
Options
	:channels - A list of channels to subscribe this matcher to
	:once? - If true, the matcher will be automatically unmatched after the first match (default: false)

Examples
Basic Usage
Antenna.match(Antenna, %{tag: _, success: false}, fn channel, message ->
  Logger.warning("The processing failed for [" <> 
    inspect(channel) <> "], result: " <> inspect(message))
end, channels: [:rabbit])
One-time Matcher
# This matcher will only trigger once and then be removed
Antenna.match(Antenna, {:init, pid}, fn _, {:init, pid} ->
  send(pid, :initialized)
end, channels: [:system], once?: true)
Using Pattern Guards
Antenna.match(Antenna, 
  {:metric, name, value} when value > 100, 
  fn channel, event ->
    Logger.info("High metric value on #{channel}: #{inspect(event)}")
  end, 
  channels: [:metrics])
Multiple Channels
Antenna.match(Antenna, 
  {:user_event, user_id, _action},
  fn channel, event ->
    # Handle user events from multiple channels
    MyApp.UserEventTracker.track(channel, event)
  end,
  channels: [:user_logins, :user_actions, :user_settings])

  



    

  
    
      
    
    
      start_link(init_arg \\ [])



        
          
        

    

  


  

      

          @spec start_link([{:id, atom()}]) :: Supervisor.on_start()


      


Starts the Antenna matcher tree for the id given. 

  



    

  
    
      
    
    
      subscribe(id \\ Antenna, channels, pid)



        
          
        

    

  


  

      

          @spec subscribe(id :: id(), channels :: channel() | [channel()] | MapSet.t(), pid()) ::
  :ok


      


Subscribes a matcher process specified by pid to a channel(s)

  



    

  
    
      
    
    
      unhandle(id \\ Antenna, handlers, pid)



        
          
        

    

  


  

      

          @spec unhandle(id :: id(), handlers :: handler() | [handler()] | MapSet.t(), pid()) ::
  :ok


      


Removes a handler from the matcher process specified by pid

  



    

  
    
      
    
    
      unmatch(id \\ Antenna, match)


        (macro)


        
          
        

    

  


  

Undeclares a matcher for tagged events previously declared with Antenna.match/4.
Accepts both an original match or a name returned by Antenna.match/4,
  which is effectively Macro.to_string(match).
Example
Antenna.unmatch(Antenna, %{tag: _, success: false})

  



    

  
    
      
    
    
      unsubscribe(id \\ Antenna, channels, pid)



        
          
        

    

  


  

      

          @spec unsubscribe(id :: id(), channels :: channel() | [channel()] | MapSet.t(), pid()) ::
  :ok


      


Unsubscribes a previously subscribed matcher process specified by pid from the channel(s)

  


        

      

      
        Functions (Internals)

        


  
    
      
    
    
      registered_matchers(id)



        
          
        

    

  


  

      

          @spec registered_matchers(id :: id()) :: %{
  required(term()) => {pid(), Supervisor.child_spec()}
}


      


Returns a map of matches to matchers

  


        

      

      
        Types

        


  
    
      
    
    
      channel()



        
          
        

    

  


  

      

          @type channel() :: atom() | term()


      


The identifier of the channel, messages can be sent to, preferrably atom()

  



  
    
      
    
    
      event()



        
          
        

    

  


  

      

          @type event() :: term()


      


The event being sent to the listeners

  



  
    
      
    
    
      handler()



        
          
        

    

  


  

      

          @type handler() ::
  (event() -> term())
  | (channel(), event() -> term())
  | Antenna.Matcher.t()
  | pid()
  | GenServer.name()


      


The actual handler to be associated with an event(s). It might be either a function
  or a process id, in which case the message of a following shape will be sent to it.
{:antenna_event, channel, event}

  



  
    
      
    
    
      id()



        
          
        

    

  


  

      

          @type id() :: module()


      


The identifier of the isolated Antenna

  



  
    
      
    
    
      matcher()



        
          
        

    

  


  

      

          @type matcher() :: term()


      


The matcher to be used for the event, e.g. {:tag_1, a, _} when is_nil(a)
The pattern matching is done on the Antenna process, so it might be
  either a function or a process id.

  



  
    
      
    
    
      opts()



        
          
        

    

  


  

      

          @type opts() :: [id: id(), channels: channel() | [channel()], once?: boolean()]


      


The options to be passed to Antenna.match/4 and Antenna.unmatch/2
	:id - the identifier of the Antenna, defaults to Antenna
	:channels - a list of channels to subscribe the matcher to
	:once? - if true, the matcher would be removed after the first match


  


        

      


  

    
Antenna.ExUnit 
    



      
The helper module for ExUnit tests. It exports assert_event/2 and refute_event/2 functions to assert and refute events.

      


      
        Summary


  
    Functions
  


    
      
        assert_event(id \\ nil, match, channels, event, timeout \\ 5000, fun)

      


        Asserts that an event is received within the specified timeout.



    





      


      
        Functions

        


    

    

  
    
      
    
    
      assert_event(id \\ nil, match, channels, event, timeout \\ 5000, fun)


        (macro)


        
          
        

    

  


  

Asserts that an event is received within the specified timeout.
Example
assert_event(Antenna, {:foo, _}, :chan_1, _, fn -> provoke_foo(Fooer) end)

  


        

      


  

    
Antenna.Matcher behaviour
    



      
The behaviour defining matchers in the Antenna pub-sub model.
While the matcher might be a simple function, or a process identifier,
  using this behaviour would have your testing process simplified.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        The module implementing handle_match/2 callback



    





  
    Callbacks
  


    
      
        handle_match(channel, event)

      


        The funcion to be called back upon match



    





  
    Functions
  


    
      
        child_spec(init_arg)

      


        Returns a specification to start this module under a supervisor.



    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: module()


      


The module implementing handle_match/2 callback

  


        

      

      
        Callbacks

        


  
    
      
    
    
      handle_match(channel, event)



        
          
        

    

  


  

      

          @callback handle_match(channel :: Antenna.channel(), event :: Antenna.event()) :: term()


      


The funcion to be called back upon match

  


        

      

      
        Functions

        


  
    
      
    
    
      child_spec(init_arg)



        
          
        

    

  


  

Returns a specification to start this module under a supervisor.
See Supervisor.

